


Introduction to linked data

# Sparql revisited



# http://training.epimorphics.com



The screenshot shows a web browser window with the URL [training.epimorphics.com](http://training.epimorphics.com) in the address bar. The page features a blue header with the text "Introduction to Linked Data" and "epimorphics.com". Below the header, there is a red sidebar menu with a blue border. The menu items are:


- ▶ Classroom – learn RDF and SPARQL by doing
- ▶ Example Linked Data
- ▶ London Museums
- ▶ London Tube Stations
- ▶ Linked data course notes
- ▶ What is linked data?
- ▶ How does linked data work?
- ▶ Examples of linked data applications
- ▶ Querying linked Data
- ▶ Getting started
- ▶ Resources
- ▶ Resources and links

# http://training.epimorphics.com/classroom

om

The screenshot shows the Epimorphics Classroom web application. The browser's address bar displays the URL `training.epimorphics.com/classroom`. The application header includes the title "Classroom" with the tagline "learning linked data", and navigation links for "RDF data editor", "SPARQL query editor" (highlighted with a red box), and "Prefixes editor". The Epimorphics Ltd logo is in the top right corner. Below the header, there are "EXAMPLE QUERIES:" buttons for "Selection of triples" and "all OWL classes". The main area is a large text editor for writing SPARQL queries, with a line number "1" visible on the left. At the bottom, there is a "SPARQL ENDPOINT" dropdown menu showing "training.epimorphics.com" (highlighted with a red box). To the left of this are buttons for "clear" and "add known prefixes" (highlighted with a red box). To the right is a "RESULTS" dropdown menu showing "plain text" (highlighted with a red box), followed by a "perform query" button. Below the endpoint menu, a "QUERY RESULTS" section shows a sample result: "London underground" (highlighted with a red box), with a tooltip indicating it is an "RDF data editor" link. The footer text reads "Linked data training resources from Epimorphics Ltd".

# Data Summary




[http://training.epimorphics.com/transport/london-underground/station/russell\\_square](http://training.epimorphics.com/transport/london-underground/station/russell_square)

[http://training.epimorphics.com/transport/london-underground/line/piccadilly\\_line](http://training.epimorphics.com/transport/london-underground/line/piccadilly_line)

[http://training.epimorphics.com/transport/london-underground/connection/piccadilly\\_line/acton\\_town/ealing\\_common](http://training.epimorphics.com/transport/london-underground/connection/piccadilly_line/acton_town/ealing_common)

[http://training.epimorphics.com/culture/london/museum/British\\_Museum](http://training.epimorphics.com/culture/london/museum/British_Museum)

# What is the URI for Russell Square



```
SELECT ?thing {  
  ?thing    rdfs:label    "Russell Square"  
}
```

1 result in 0 min 0.080 s

1	-----
2	thing
3	=====
4	lu-s:russell_square
5	-----
6	

# Exercises

- What points of interest are nearby Russell Square
- What are the names of the points of interest nearby Russell Square

# What Points of Interest are nearby Russell Square?

3 results in 0 min 0.074 s

1	-----
2	poi
3	=====
4	museum:British Museum
5	museum:Foundling Museum
6	museum:October Gallery
7	-----
8	

```
SELECT ?poi {  
  lu-s:russell_square    metro:nearbyTo    ?poi .  
}
```

# What are the names of the points of interest nearby Russell Square?

3 results in 0 min 0.142 s

1	- - - - -
2	poiName
3	=====
4	"British Museum"
5	"Foundling Museum"
6	"October Gallery"
7	- - - - -
8	

```
SELECT ?poiName {  
  lu-s:russell_square    metro:nearbyTo    ?poi .  
  ?poi                   rdfs:label        ?poiName .  
}
```



# Language Tags

Find all the properties of the thing whose name in English is "Underground Station"

```
SELECT ?thing ?predicate ?object {  
  ?thing    rdfs:label    "Underground station"@en  
  ;        ?predicate    ?object  
  .  
}
```

7 results in 0 min 0.133 s

1	-----		
2	thing	predicate	object
3	=====	=====	=====
4	metro:UndergroundStation	rdf:type	owl:Class
5	metro:UndergroundStation	rdfs:label	"Underground station"@en
6	metro:UndergroundStation	rdfs:subClassOf	<http://schema.org/Place>
7	metro:UndergroundStation	rdfs:subClassOf	owl:Thing
8	metro:UndergroundStation	rdfs:subClassOf	<http://training.epimorphics.c
9	metro:UndergroundStation	rdfs:subClassOf	:b0
10	metro:UndergroundStation	rdfs:subClassOf	:b1
11	-----		

# Datatypes

- Write an integer as:

`"10"^^xsd:int`


- General form is:

`"{lexical form}"^^{uri of the datatype}`

- Short forms:

- 10 is `"10"^^xsd:integer`
- 10.5 is `"10.5"^^xsd:decimal`
- 10e2 is `"10e2"^^xsd:double`
- true is `"true"^^xsd:boolean`

# What lines is each station on?



```
SELECT ?stationName ?lineName
WHERE {
```

?station	rdf:type	metro:UndergroundStation
	; rdfs:label	?stationName
	; metro:incomingConnection	?con .
?con	metro:onLine	?line .
?line	rdfs:label	?lineName .

# Should get something like this


820 results in 0 min 0.179 s

	stationName	Duplicates!	lineName
1			
2			
3			
4	"Acton Town"		"District Line"
5	"Acton Town"		"District Line"
6	"Acton Town"		"Piccadilly Line"
7	"Acton Town"		"Piccadilly Line"
8	"Acton Town"		"Piccadilly Line"
9	"Aldgate"		"Circle Line"
10	"Aldgate"		"Circle Line"
11	"Aldgate"		"Metropolitan Line"
12	"Aldgate East"		"District Line"
13	"Aldgate East"		"District Line"
14	"Aldgate East"		"Hammersmith & City Line"
15	"Aldgate East"		"Hammersmith & City Line"
16	"Alperton"		"Piccadilly Line"
17	"Alperton"		"Piccadilly Line"
18	"Amersham"		"Metropolitan Line"
19	"Angel"		"Northern Line"
20	"Angel"		"Northern Line"
21	"Archway"		"Northern Line"
22	"Archway"		"Northern Line"
23	"Arnos Grove"		"Piccadilly Line"
24	"Arnos Grove"		"Piccadilly Line"
25	"Arsenal"		"Piccadilly Line"


# Why did we get duplicates?

?station metro:incomingConnection  
?con metro:onLine

?con .  
?line .



# Eliminating Duplicates



**SELECT DISTINCT** ?stationName ?lineName

**WHERE** {

?station	rdf:type	metro:UndergroundStation
	rdfs:label	?stationName
	metro:incomingConnection	?conn .
?conn	metro:onLine	?line .
?line	rdfs:label	?lineName .

# Duplicates have gone

417 results in 0 min 0.181 s

1	-----		
2	stationName	lineName	
3	=====	=====	
4	"Acton Town"	"District Line"	
5	"Acton Town"	"Piccadilly Line"	
6	"Aldgate"	"Circle Line"	
7	"Aldgate"	"Metropolitan Line"	
8	"Aldgate East"	"District Line"	
9	"Aldgate East"	"Hammersmith & City Line"	
10	"Alperton"	"Piccadilly Line"	
11	"Amersham"	"Metropolitan Line"	
12	"Angel"	"Northern Line"	
13	"Archway"	"Northern Line"	
14	"Arnos Grove"	"Piccadilly Line"	
15	"Arsenal"	"Piccadilly Line"	
16	"Baker Street"	"Bakerloo Line"	
17	"Baker Street"	"Circle Line"	
18	"Baker Street"	"Hammersmith & City Line"	
19	"Baker Street"	"Jubilee Line"	
20	"Baker Street"	"Metropolitan Line"	
21	"Balham"	"Northern Line"	
22	"Bank"	"Waterloo & City Line"	
23	"Bank"	"Central Line"	
24	"Bank"	"Docklands Light Railway"	
25	"Bank"	"Northern Line"	

Can also use **REDUCED** instead of **DISTINCT**

# Sorting - Order By


```
SELECT DISTINCT ?stationName ?lineName
WHERE {
    ?station    rdf:type        metro:UndergroundStation
                ;    rdfs:label    ?stationName
                ;    metro:incomingConnection ?conn
                .
    ?conn        metro:onLine    ?line .
    ?line        rdfs:label      ?lineName .
}
ORDER BY ?lineName ?stationName
# ORDER BY DESC(?lineName) DESC(?stationName)
```



# Exercise

- Find all the museums and order by distance from the nearest tube station
  - hint: `rdf:type <http://schema.org/Museum>`
- Extra Credit:
  - List all the Bakerloo line stations from North to South
    - hint `geo:lat`

# http://training.epimorphics.com



The screenshot shows a web browser window with the URL [training.epimorphics.com](http://training.epimorphics.com). The page features a blue header with the text "Introduction to Linked Data" and "epimorphics.com". Below the header, there is a red sidebar menu with the following items:

- ▶ Classroom – learn RDF and SPARQL by doing
- ▶ Example Linked Data
- ▶ London Museums
- ▶ London Tube Stations
- ▶ Linked data course notes
- ▶ What is linked data?
- ▶ How does linked data work?
- ▶ Examples of linked data applications
- ▶ Querying linked Data
- ▶ Getting started
- ▶ Resources
- ▶ Resources and links

# Find all the museums and order by distance from the nearest tube station

```
SELECT DISTINCT ?museumName ?distance
WHERE {
    ?musuem    a <http://schema.org/Museum>
    ;          rdfs:label      ?museumName
    ;          metro:nearestUndergroundStationDistance      ?distance.
} ORDER BY ?distance
```

24 results in 0 min 0.101 s

1	-----	
2	museumName	distance
3	-----	-----
4	"Wellcome Collection"	143
5	"Whitechapel Gallery"	160
6	"The Photographers' Gallery"	181
7	"Russian Cultural Centre (London)"	232
8	"Whitfield Fine Art"	256
9	"Saatchi Gallery"	260
10	"October Gallery"	304
11	"Victoria and Albert Museum"	313
12	"Foundling Museum"	316
13	"Art in Perpetuity Trust"	327
14	"National Maritime Museum"	331
15	"Queen's Gallery"	400
16	"British Museum"	431
17	"Institute of Contemporary Arts"	433
18	"Tate Britain"	461
19	"Sartorial Contemporary Art"	491
20	"Tate Modern"	543
21	"Parasol unit foundation for contemporary art"	687
22	"White Cube"	688
23	"De Morgan Centre"	754
24	"Serpentine Gallery"	806
25	"William Morris Gallery"	920
26	"Geffrye Museum"	977
27	"Dulwich Picture Gallery"	2684
28	-----	-----

# FILTERS: only show the ones within 500M

```
SELECT DISTINCT ?museumName ?distance
WHERE {
    ?musuem    a <http://schema.org/Museum>
    ;          rdfs:label      ?museumName
    ;          metro:nearestUndergroundStationDistance    ?distance.
    FILTER ( ?distance <= 500 )
} ORDER BY ?distance
```

16 results in 0 min 0.100 s

	-----	
	museumName	distance
	=====	=====
1		
2		
3	-----	-----
4	"Wellcome Collection"	143
5	"Whitechapel Gallery"	160
6	"The Photographers' Gallery"	181
7	"Russian Cultural Centre (London)"	232
8	"Whitfield Fine Art"	256
9	"Saatchi Gallery"	260
10	"October Gallery"	304
11	"Victoria and Albert Museum"	313
12	"Foundling Museum"	316
13	"Art in Perpetuity Trust"	327
14	"National Maritime Museum"	331
15	"Queen's Gallery"	400
16	"British Museum"	431
17	"Institute of Contemporary Arts"	433
18	"Tate Britain"	461
19	"Sartorial Contemporary Art"	491
20	-----	-----
21		

# LIMIT: show only 10 results

```
SELECT DISTINCT ?museumName ?distance
WHERE {
    ?musuem    a <http://schema.org/Museum>
    ;          rdfs:label      ?museumName
    ;          metro:nearestUndergroundStationDistance    ?distance.
    FILTER ( ?distance <= 500 )
} ORDER BY ?distance
LIMIT 10
```

10 results in 0 min 0.098 s

	museumName	distance
1		
2		
3		
4	"Wellcome Collection"	143
5	"Whitechapel Gallery"	160
6	"The Photographers' Gallery"	181
7	"Russian Cultural Centre (London)"	232
8	"Whitfield Fine Art"	256
9	"Saatchi Gallery"	260
10	"October Gallery"	304
11	"Victoria and Albert Museum"	313
12	"Foundling Museum"	316
13	"Art in Perpetuity Trust"	327
14		
15		

# OFFSET: start at result 10

```
SELECT DISTINCT ?museumName ?distance
WHERE {
    ?musuem    a <http://schema.org/Museum>
    ;          rdfs:label      ?museumName
    ;          metro:nearestUndergroundStationDistance      ?distance.
    FILTER ( ?distance <= 500 )
} ORDER BY ?distance
LIMIT 10
OFFSET 10
```

6 results in 0 min 0.101 s

1	-----	
2	museumName	distance
3	-----	-----
4	"National Maritime Museum"	331
5	"Queen's Gallery"	400
6	"British Museum"	431
7	"Institute of Contemporary Arts"	433
8	"Tate Britain"	461
9	"Sartorial Contemporary Art"	491
10	-----	-----
11		

# List the Bakerloo line stations from north to south

<http://www.w3.org/TR/sparql11-query/>

25 results in 0 min 0.161 s

1	-----	
2	stationName	lat
3	=====	
4	"Harrow & Wealdstone"	51.5925
5	"Kenton"	51.5816
6	"South Kenton"	51.5701
7	"North Wembley"	51.5621
8	"Wembley Central"	51.5519
9	"Stonebridge Park"	51.5439
10	"Harlesden"	51.5362
11	"Kilburn Park"	51.5351
12	"Queen's Park"	51.5341
13	"Willesden Junction"	51.5326
14	"Kensal Green"	51.5304
15	"Maida Vale"	51.53
16	"Warwick Avenue"	51.5235
17	"Regent's Park"	51.5234
18	"Baker Street"	51.5226
19	"Marylebone"	51.5225
20	"Edgware Road (Bak)"	51.5199
21	"Paddington"	51.5154
22	"Oxford Circus"	51.515
23	"Piccadilly Circus"	51.5098
24	"Charing Cross"	51.508
25	"Embankment"	51.5074

geo:lat, geo:long

metro:

UndergroundStation

# List the Bakerloo line stations from north to south

```
SELECT DISTINCT ?stationName ?lat
WHERE {
    ?station rdf:type metro:UndergroundStation
        ; rdfs:label ?stationName
        ; metro:incomingConnection/metro:onLine
            lu-line:bakerloo_line
        ; geo:lat ?lat .
}
order by desc(?lat)
```



# List all the stations and their nearby points of interest

```
SELECT DISTINCT ?stationName ?poiName {  
    ?station rdf:type metro:UndergroundStation  
        ; rdfs:label ?stationName  
        ; metro:nearbyTo ?poi  
    .  
    ?poi rdfs:label ?poiName .  
} ORDER BY ?stationName
```

# We Get

24 results in 0 min 0.109 s


2	stationName	poiName
3		
4	"Aldgate East"	"Whitechapel Gallery"
5	"Angel"	"Sartorial Contemporary Art"
6	"Blackfriars"	"Tate Modern"
7	"Bond Street"	"Whitfield Fine Art"
8	"Brixton"	"Dulwich Picture Gallery"
9	"Cutty Sark"	"National Maritime Museum"
10	"Deptford Bridge"	"Art in Perpetuity Trust"
11	"East Putney"	"De Morgan Centre"
12	"Euston Square"	"Wellcome Collection"
13	"Holborn"	"Russian Cultural Centre (London)"
14	"Lancaster Gate"	"Serpentine Gallery"
15	"London Bridge"	"White Cube"
16	"Old Street"	"Geffrye Museum"
17	"Old Street"	"Parasol unit foundation for contemporary art"
18	"Oxford Circus"	"The Photographers' Gallery"
19	"Piccadilly Circus"	"Institute of Contemporary Arts"
20	"Pimlico"	"Tate Britain"
21	"Russell Square"	"British Museum"
22	"Russell Square"	"Foundling Museum"
23	"Russell Square"	"October Gallery"
24	"Sloane Square"	"Saatchi Gallery"
25	"South Kensington"	"Victoria and Albert Museum"
26	"Victoria"	"Queen's Gallery"
27	"Walthamstow Central"	"William Morris Gallery"
28		

That is not enough stations!

# Using OPTIONAL to List all the stations and their nearby points of interest

```
SELECT DISTINCT ?stationName ?poiName
WHERE {
    ?station rdf:type metro:UndergroundStation
             ; rdfs:label ?stationName .
    OPTIONAL {
        ?station metro:nearbyTo ?poi .
        ?poi rdfs:label ?poiName .
    }
} ORDER BY ?stationName
```

# Now we Get

 309 results in 0 min 0.176 s

	stationName	poiName
1		
2		
3		
4	"Acton Town"	
5	"Aldgate"	
6	"Aldgate East"	"Whitechapel Gallery"
7	"All Saints"	
8	"Alperton"	
9	"Amersham"	
10	"Angel"	"Sartorial Contemporary Art"
11	"Archway"	
12	"Arnos Grove"	
13	"Arsenal"	
14	"Baker Street"	
15	"Balham"	
16	"Bank"	
17	"Barbican"	
18	"Barking"	
19	"Barkingside"	
20	"Barons Court"	
21	"Bayswater"	
22	"Beckton"	
23	"Beckton Park"	
24	"Becontree"	
25	"Belsize Park"	
26	"Bermondsey"	
27	"Bethnal Green"	
28	"Blackfriars"	"Tate Modern"

# UNION: Find the names of all lines and the names of all stations

```
SELECT DISTINCT ?thing ?name
WHERE {
  {
    ?thing rdf:type metro:UndergroundStation .
  } UNION {
    ?thing rdf:type metro:UndergroundLine .
  }
  ?thing rdfs:label ?name
    ;    rdf:type    ?type
} ORDER BY ?type
```

319 results in 0 min 0.131 s

1	-----	-----
2	thing	name
3	-----	-----
4	lu-l:bakerloo line	"Bakerloo Line"
5	lu-l:central line	"Central Line"
6	lu-l:circle line	"Circle Line"
7	lu-l:district line	"District Line"
8	lu-l:docklands light railway	"Docklands Light Railway"
9	lu-l:east london line	"East London Line"
10	lu-l:hammersmith_and_city_line	"Hammersmith & City Line"
11	lu-l:jubilee line	"Jubilee Line"
12	lu-l:metropolitan line	"Metropolitan Line"
13	lu-l:northern line	"Northern Line"
14	lu-l:piccadilly line	"Piccadilly Line"
15	lu-l:victoria line	"Victoria Line"
16	lu-l:waterloo_and_city_line	"Waterloo & City Line"
17	lu-s:acton town	"Acton Town"
18	lu-s:aldgate	"Aldgate"
19	lu-s:aldgate east	"Aldgate East"
20	lu-s:all saints	"All Saints"
21	lu-s:alperton	"Alperton"
22	lu-s:amersham	"Amersham"
23	lu-s:angel	"Angel"
24	lu-s:archway	"Archway"
25	lu-s:arnos grove	"Arnos Grove"

# Aggregation - Count the stations

```
SELECT (COUNT( DISTINCT ?station) AS ?count)
WHERE {
    ?station rdf:type metro:UndergroundStation
}
```

1 result in 0 min 0.162 s

1	-----
2	count
3	=====
4	306
5	-----
6	

# Count the stations on each line

```
SELECT ?lineName (COUNT(DISTINCT ?station) AS ?numStations)
WHERE {
    ?station rdf:type metro:UndergroundStation
            ; rdfs:label ?stationName
            ; metro:incomingConnection/metro:onLine ?line .
    ?line rdfs:label ?lineName .
}
GROUP BY ?lineName
ORDER BY ?numStations
```

# Should get something like this

13 results in 0 min 0.174 s

1	-----		
2	lineName	numStations	
3	=====	=====	
4	"Waterloo & City Line"	2	
5	"East London Line"	9	
6	"Victoria Line"	16	
7	"Bakerloo Line"	25	
8	"Circle Line"	27	
9	"Jubilee Line"	27	
10	"Hammersmith & City Line"	28	
11	"Metropolitan Line"	34	
12	"Docklands Light Railway"	38	
13	"Central Line"	49	
14	"Northern Line"	50	
15	"Piccadilly Line"	52	
16	"District Line"	60	
17	-----		



# Aggregate Functions

- COUNT(?var)
- SUM(?var)
- AVG(?var)
- MIN(?var)
- MAX(?var)
- GROUP\_CONCAT(?var; separator=" ")
- SAMPLE(?var)

# Exercises

- Count all the stations on the Bakerloo line
- Which station has the most nearby POI
- Extra credit
  - List all the stations and their nearby points of interest, one line per station

# Count all the stations on the Bakerloo line

```
SELECT (COUNT(DISTINCT ?station) AS ?numStations)
WHERE {
    ?station rdf:type metro:UndergroundStation
        ;      metro:incomingConnection/metro:onLine
                lu-l:bakerloo_line
        .
}
```

1 result in 0 min 0.065 s

1	
2	numStations
3	=====
4	25
5	
6	

# Which station has the most nearby POI

```
SELECT ?stationName
      (COUNT( DISTINCT ?poi) AS ?count)
WHERE {
    ?station rdf:type metro:UndergroundStation
              ; rdfs:label ?stationName
              ; metro:nearbyTo ?poi
              .
} GROUP BY ?stationName
   ORDER BY DESC(?count)
# LIMIT 1
```

21 results in 0 min 0.064 s

1	-----	
2	stationName	count
3	=====	=====
4	"Russell Square"	3
5	"Old Street"	2
6	"Aldgate East"	1
7	"Angel"	1
8	"Blackfriars"	1
9	"Bond Street"	1
10	"Brixton"	1
11	"Cutty Sark"	1
12	"Deptford Bridge"	1
13	"East Putney"	1
14	"Euston Square"	1
15	"Holborn"	1
16	"Lancaster Gate"	1
17	"London Bridge"	1
18	"Oxford Circus"	1
19	"Piccadilly Circus"	1
20	"Pimlico"	1
21	"Sloane Square"	1
22	"South Kensington"	1
23	"Victoria"	1
24	"Walthamstow Central"	1
25	-----	-----

# Federated Query

```
SELECT ?label ?lang ?comment {  
  ?station rdfs:label "Russell Square"  
    ;      metro:nearbyTo ?poi .  
  ?poi rdfs:label ?label  
    ; owl:sameAs ?poiDBP .  
  
  SERVICE <http://dbpedia.org/sparql> {  
    ?poiDBP rdfs:comment ?comment .  
    BIND(LANG(?comment) AS ?lang)  
    FILTER(?lang = 'fr' || ?lang = 'de' || ?lang = 'es')  
  }  
}
```

# Datasets and Named Graphs

```
SELECT * {  
  GRAPH <http://example.org/graph> {  
    ?s ?p ?o  
  }  
}
```

# And you should get something like

4 results in 0 min 0.289 s

```
1 -----
2 | label                | lang | comment
3 =====
4 | "British Museum"    | "de" | "Das British Museum (BM; de
5 | "British Museum"    | "es" | "El Museo Británico (en ing
6 | "British Museum"    | "fr" | "Le British Museum (qui peu
7 | "Foundling Museum"  | "es" | "El Museo Foundling de Lond
8 -----
9
```

# ASK

```
ASK {  
  lu-s:oxford_circus ?p ?o .  
}
```

Response in 0 min 0.062 s

1 yes



# DESCRIBE

DESCRIBE lu-s:oxford\_circus

```
18
19 lu-station:oxford_circus
20     a                transport:UndergroundStation ;
21     rdfs:label        "Oxford Circus" ;
22     transport:incomingConnection <http://training.epimorphics.com/transport/london-underground/connection/central_line/tottenham_court_
23     transport:nearbyTo <http://training.epimorphics.com/culture/london/museum/The_Photographers'_Gallery> ;
24     transport:outgoingConnection <http://training.epimorphics.com/transport/london-underground/connection/victoria_line/oxford_circus/g
25     geo:lat            51.515 ;
26     geo:long           -0.1415 .
27
```

# CONSTRUCT

```
CONSTRUCT {  
  ?station metro:nameOfNearbyPOI ?poiName .  
} WHERE {  
  ?station a metro:UndergroundStation ;  
           metro:nearbyTo/rdfs:label ?poiName  
}
```

```
17 @prefix rdf:    <http://www.w3.org/1999/02/22-rdf-syntax-ns#> .  
18  
19 lu-station:victoria transport:nameOfNearbyPOI  
20     "Queen's Gallery" .  
21  
22 lu-station:blackfriars  
23     transport:nameOfNearbyPOI "Tate Modern" .  
24  
25 lu-station:south_kensington  
26     transport:nameOfNearbyPOI "Victoria and Albert Museum" .  
27  
28 lu-station:piccadilly_circus  
29     transport:nameOfNearbyPOI "Institute of Contemporary Arts" .  
30  
31 lu-station:russell_square  
32     transport:nameOfNearbyPOI "October Gallery" , "Foundling Museum" , "British Museum" .  
33
```

# UPDATE

```
INSERT {  
  ?station metro:nameOfNearbyPOI ?poiName .  
} WHERE {  
  ?station a metro:UndergroundStation ;  
           metro:nearbyTo/rdfs:label ?poiName  
}
```

# Extension Points - Magic Properties

```
SELECT DISTINCT ?poiName {  
  ?station metro:nearbyTo ?poi .  
  ?poi      rdfs:label      ?poiName .  
  FILTER( CONTAINS(LCASE(?poiName), "gallery") )  
}
```

- A typical implementation will retrieve all the labels and run the CONTAINS function on them one at a time
- Ok for our London Tube data – not so good if you have a lot of data e.g. the addresses of all houses in the UK

```
SELECT DISTINCT ?poiName {  
  ?poiName text:query "gallery" .  
  ?station metro:nearbyTo ?poi .  
  ?poi      rdfs:label      ?poiName .  
}
```

# SPARQL

- Graph patterns
- Filters
- Union and Optional
- Solution Modifiers
- Aggregation
- Datasets and named graphs
- Sub queries
- ASK, DESCRIBE, CONSTRUCT
- UPDATE
- Query federation

<http://www.w3.org/TR/sparql11-query/>


Questions?



[www.epimorphics.com](http://www.epimorphics.com)



# List all the stations and their nearby points of interest, one line per station

 20 results in 0 min 0.100 s

stationName	POIs
"Aldgate East"	"Whitechapel Gallery"
"Angel"	"Sartorial Contemporary Art"
"Blackfriars"	"Tate Modern"
"Bond Street"	"Whitfield Fine Art"
"Brixton"	"Dulwich Picture Gallery"
"Cutty Sark"	"National Maritime Museum"
"Deptford Bridge"	"Art in Perpetuity Trust"
"East Putney"	"De Morgan Centre"
"Euston Square"	"Wellcome Collection"
"Holborn"	"Russian Cultural Centre (London)"
"Lancaster Gate"	"Serpentine Gallery"
"London Bridge"	"White Cube"
"Old Street"	"Geffrye Museum   Parasol unit foundation for contemporary art"
"Oxford Circus"	"The Photographers' Gallery"
"Pimlico"	"Tate Britain"
"Russell Square"	"British Museum   Foundling Museum   October Gallery"
"Sloane Square"	"Saatchi Gallery"
"South Kensington"	"Victoria and Albert Museum"
"Victoria"	"Queen's Gallery"
"Walthamstow Central"	"William Morris Gallery"



# List all the stations and their nearby points of interest, one line per station

```
SELECT ?stationName
      (COUNT( DISTINCT ?poi) AS ?count)
WHERE {
    ?station rdf:type metro:UndergroundStation
              ; rdfs:label ?stationName
              ; metro:nearbyTo ?poi
              .
} GROUP BY ?stationName
ORDER BY ?stationName
# LIMIT 1
```

# Count the connections between each pair of lines

56 results in 0 min 0.259 s

1	-----	-----	-----
2	line1Name	line2Name	count
3	-----	-----	-----
4	"Circle Line"	"District Line"	18
5	"District Line"	"Hammersmith & City Line"	14
6	"Circle Line"	"Hammersmith & City Line"	10
7	"Circle Line"	"Metropolitan Line"	9
8	"District Line"	"Piccadilly Line"	8
9	"Hammersmith & City Line"	"Metropolitan Line"	8
10	"Metropolitan Line"	"Piccadilly Line"	8
11	"Bakerloo Line"	"Northern Line"	4
12	"Northern Line"	"Victoria Line"	4
13	"Bakerloo Line"	"Circle Line"	3
14	"Central Line"	"District Line"	3
15	"Circle Line"	"Northern Line"	3
16	"Circle Line"	"Piccadilly Line"	3
17	"Docklands Light Railway"	"Jubilee Line"	3
18	"Jubilee Line"	"Metropolitan Line"	3
19	"Piccadilly Line"	"Victoria Line"	3
20	"Bakerloo Line"	"District Line"	2
21	"Bakerloo Line"	"Hammersmith & City Line"	2
22	"Bakerloo Line"	"Jubilee Line"	2
23	"Central Line"	"Circle Line"	2
24	"Central Line"	"Docklands Light Railway"	2
25	"Central Line"	"Hammersmith & City Line"	2
26	"Central Line"	"Jubilee Line"	2
27	"Central Line"	"Northern Line"	2
28	"Circle Line"	"Jubilee Line"	2

# Count the connections between each pair of lines

```
SELECT DISTINCT ?line1Name ?line2Name
              (COUNT(DISTINCT ?station) AS ?count)
WHERE {
    ?station rdf:type metro:UndergroundStation
        ; rdfs:label ?stationName
        ; metro:incomingConnection/metro:onLine ?line1
        ; metro:incomingConnection/metro:onLine ?line2
        .
    FILTER( ?line1 != ?line2 )
    ?line1 rdfs:label ?line1Name .
    ?line2 rdfs:label ?line2Name .
    FILTER (?line1Name < ?line2Name)
}
GROUP BY ?line1Name ?line2Name
ORDER BY DESC(?cc int)
```